

## Laboratorio di Architetture degli elaboratori n. 17, 16-17/5/02

### Esercizio 1.

Si progetti una subroutine assembler 68000 che accetti in ingresso:

1. l'indirizzo base di una lista  $L$  di numeri interi a 16 bit nel registro A0,
2. un numero positivo  $n$  di 16 bit nella word meno significativa del registro D0,
3. l'indirizzo base di un blocco di 3 word nel registro A1.

La subroutine deve copiare  $n$  nel campo dati del blocco di indirizzo A1 e inserire il blocco di indirizzo A1 come ultimo elemento della lista  $L$ .

Ad esempio, se  $L = 3,6,7,8,11$  e  $n = 9$  la subroutine deve costruire la lista  $L = 3,6,7,8,11,9$ .

```
tailInsert  movem.l  a0/a3,-(sp) ASSUMENDO L non nulla
            clr.l    (a1)
            move.w   d0,4(a1)
            move.l   a0,a2
stopOnLast  cmpi.l   #0,(a2)
            beq.s   stopped
            move.l   (a2),a2
            bra     stopOnLast
stopped     move.l   a1,(a2)      A2 si ferma sull'ultimo nodo
exit       movem.l  (sp)+,a0/a3
            rts

tailInsert  movem.l  a2/d1,-(sp) senza assunzioni su L
            clr.l    (a1)
            move.w   d0,4(a1)
            cmpa.l   #0,a0
            beq.s   emptyapp
            move.l   a0,a2
stopOnLast  cmpi.l   #0,(a2)
            beq.s   stopped      A2 si ferma sull'ultimo nodo
            move.l   (a2),a2
            bra     stopOnLast
stopped     move.l   a1,(a2)
            bra     exit
emptyapp    move.l   a1,a0
exit       movem.l  (sp)+,a2/d1
            rts
```

### Esercizio 2.

Si progetti una subroutine assembler 68000 che accetti in ingresso:

1. l'indirizzo base di una lista  $L$  di numeri interi a 16 bit nel registro A0,
2. un numero positivo  $n$  di 16 bit nella word meno significativa del registro D0.

La subroutine deve cancellare dalla lista  $L$  tutte le occorrenze del numero  $n$ . Ad esempio, se  $L = 0,1,5,3,5,5,4$  e  $n = 5$  la subroutine deve produrre la lista  $L = 0,1,3,4$ .

```

removeN    movem.l  a1/a2,-(sp) senza assunzioni su L
           movea.l  #0,a1
           movea.l  a0,a2
loop       cmpa.l   #0,a2
           beq.s    exit
           cmp.w    4(a2),d0
           bne.s    continue
           cmpa.l   #0,a1
           beq.s    headdete
           move.l   (a2),(a1)
           bra      next
headdete   movea.l  (a2),a0
           bra      next
continue   movea.l  a2,a1
next       movea.l  (a2),a2
           bra      loop
exit       movem.l  (sp)+,a1/a2
           rts

```